
	Reason for change:
	1. TTCN code of test case 15.8 shall be re-used for WLAN/FBBA test cases G.15.7 and H.15.7 
2. OptionTag “precondion” is not included for C.11c; OptionTag “timer” is not included in A.2.9 at all
3. Common function for SDP message of C.11x step 4/8 shall support C.11c 

4. Common function needed for Call-Hold test cases in WLAN

5. Common MO call establishment has been implemented for EUTRA only

6. f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9 has been implemented for EUTRA only
7. f_IMS_MOCallwithOIR has been implemented for EUTRA only
8. f_IMS_MTCallwithTIR has been implemented for EUTRA only
9. f_TC_15_21a_TC_15_21c_Body has been implemented for EUTRA only
10. Test bodies of several further EUTRA/WLAN/FBBA/CatM1 test cases have been IPCAN specific even though use of common test bodies is possible

	
	

	Summary of change:
	1. Split of fl_TC_15_8_Steps10_17
2. p_AdditionalOptionTagsForSupported re-introduced for f_IMS_InviteRequest_MessageHeaderTX; OptionTag “timer” removed from Supported header in f_IMS_InviteRequest_MessageHeaderTX
3. f_IMS_BuildSDP_MTSpeechCall_Step4or8 enhanced to support SDP message without preconditions
4. Bearer specific steps moved from f_IMS_MO_CallOnHoldWithoutAnnouncement and f_IMS_MT_CallOnHoldWithoutAnnouncement into the test case specific implementation to allow reuse in WLAN test cases
5. Enhancement of f_IMS_CC_MO_SpeechCall to support other IPCANs
6. Enhancement of f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9 to support WLAN
7. Enhancement of f_IMS_MOCallwithOIR to support other IPCANs
8. Enhancement of f_IMS_MTCallwithTIRto support other IPCANs

9. Common function f_IMS_ThreeWaySession_Creation to support three way session creation for EUTRA and WLAN
10. Use of common test bodies where possible

	Other comments:
	1. This CR does not show all details of changes made for IMS cleanup and harmonisation of the different IPCANs. Its purpose is showing the major changes and common principles.
2. Prose CR is needed for C.10 and C.38 as test procedure for MTSI conference creation differs for EUTRA and WLAN


Change 1 – Split of fl_TC_15_8_Steps10_17 
In order to re-use TTCN code of test case 15.8 for WLAN/FBBA test cases G.15.7 and H.15.7 fl_TC_15_8_Steps10_17 has been split and renamed to fl_IMS_MO_ComunicationForwardingOnNonReply_C21C21b_Steps10_17:
	  function f_IMS_MO_ComunicationForwardingOnNonReply_ModifyDialog(integer p_CseqValue) runs on IMS_PTC 

  {

    // set new tag for the dialog:

    f_IMS_PTC_ImsInfo_DialogSetLocalTag(f_IMS_GenerateTag("INVITE-FORWARDED"));     /* @sic R5s140621 1st change: in terms of the response the To-tag in the local tag here sic@ */

    // cseq needs to be reset to the value of the INVITE (@sic R5s140621 2nd change sic@)

    f_IMS_PTC_ImsInfo_CseqSet(dialogRemote, p_CseqValue);

    // 15.8 Step 11 Send 183 Session Progress

    f_IMS_PTC_ImsInfo_DialogSetSessIdTX("22222222");       /* @sic R5-150707 sic@ */

    f_IMS_PTC_ImsInfo_DialogSetSessVersionTX(22222221);    /* @sic R5-150707 (NOTE: session version is incremented before being used in SDP message) sic@ */

    f_IMS_PTC_ImsInfo_DialogSetPrevSdpMessageRX(omit);     /* @sic R5s150666, R5-153525 sic@ */
  }

	  function f_IMS_MO_ComunicationForwardingOnNonReply_ReceivePRACK_Send200OK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                                            template (value) MessageHeader p_MessageHeader_PrevReliableResponse,

                                                                            charstring p_ForwardedToCalleeContactUri) runs on IMS_PTC

  {

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var template (present) MessageHeader v_MessageHeader_Request;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var PRACK_Request v_PrackRequest;

    v_MessageHeader_Request := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, p_MessageHeader_PrevReliableResponse, omit);

    v_IMS_DATA_REQ := f_IMS_ReceiveRequest(car_IMS_Prack_Request(cr_PRACK_Request(f_SIP_BuildSipUri_RX(p_ForwardedToCalleeContactUri), v_MessageHeader_Request)));

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader))));

  }

	  function f_IMS_MO_ComunicationForwardingOnNonReply_Send200OK_ReceiveACK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                                          charstring p_ForwardedToCalleeContactUri,

                                                                          template (value) HistoryInfo p_HistoryInfo) runs on IMS_PTC

  {

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, -, p_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := p_HistoryInfo;

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response)));

    v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(f_SIP_BuildSipUri_RX(p_ForwardedToCalleeContactUri), v_MessageHeader_Request)));

  }

	  function fl_IMS_MO_ComunicationForwardingOnNonReply_C21C21b_Steps10_17(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                 SDP_Message p_PreviousSDP_Message) runs on IMS_PTC return template (present) SipUrl

  { /* return contact of the callee to which the call is forwarded to */

    /* @sic R5s140905: new parameter p_PreviousSDP_Message sic@ */

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageStep2 := p_InviteRequestWithSdp.SdpOffer;

    var charstring v_ForwardedToCalleeContactUri := "sip:15.8@" & tsc_IMS_CallerDomain;


    var template (value) HistoryInfo v_HistoryInfo := cs_HistoryInfo(v_TargetedTo);

    var template (value) SipUrl v_TargetedTo := f_SIP_BuildSipUri_TX("sip:user@company.com");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;

    var PRACK_Request v_PrackRequest;

    // 15.8 Step 10 Send 181 Call is being forwarded

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine181, f_IMS_InviteResponse_181_MessageHeaderTX(v_InviteRequest, v_TargetedTo))));

    f_IMS_MO_ComunicationForwardingOnNonReply_ModifyDialog(v_InviteRequest.msgHeader.cSeq.seqNumber);









    v_MessageHeader_Response := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, v_ForwardedToCalleeContactUri);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183,

                                                                   v_MessageHeader_Response,

                                                                   cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21_Step4(v_SDP_MessageStep2)))));   /* @sic R5-144402 sic@ */

    // 15.8 Step 12 - 13B (same as steps 5 - 8 of C.21)

    f_IMS_MOCallSetup_AnnexC21_Steps5_8(p_InviteRequestWithSdp, v_MessageHeader_Response, -, p_PreviousSDP_Message);  /* the procedure uses the (remote) contact address as in the 183 response (v_MessageHeader_Response)

                                                                                                                         => in fact v_ForwardedToCalleeContactUri is used

                                                                                                                         @sic R5s140905: p_PreviousSDP_Message sic@ */

    // 15.8 Step 14. 180 Ringing

    v_MessageHeader_Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, v_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := v_HistoryInfo;

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_Response)));

    // 15.8 Step 14A. Receive PRACK; 15.8 Step 14B. Send 200 OK for PRACK

    f_IMS_MO_ComunicationForwardingOnNonReply_ReceivePRACK_Send200OK(p_InviteRequestWithSdp, v_MessageHeader_Response, v_ForwardedToCalleeContactUri);







    // Step 15. Send 200 OK for INVITE; Step 15. UE acknowledges call setup

    f_IMS_MO_ComunicationForwardingOnNonReply_Send200OK_ReceiveACK(p_InviteRequestWithSdp, v_ForwardedToCalleeContactUri, v_HistoryInfo);









    return f_SIP_BuildSipUri_RX(v_ForwardedToCalleeContactUri);
  }


Change 2 – Corrections of option tags in Supported header of f_IMS_InviteRequest_MessageHeaderTX
· OptionTag “precondion” is not included for C.11c

· OptionTag “timer” is not included in A.2.9 at all
	  function f_IMS_InviteRequest_MessageHeaderTX(OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_precondition,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                               boolean p_MTSI := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_Video := pc_IMS_Video_FeatureTag,

                                               integer p_CSeqValue := 4711) runs on IMS_PTC return template (value) MessageHeader

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5-140915 MT call hold requires re-INVITE => new parameter p_Context for A4/A5 sic@ */

    /* @sic R5s140385 change 6: new parameter p_SrvccAlertingSupported sic@ */

    /* @sic R5s150086 - p_SessionIdSupported: pc_IMS_SessionIdSupported per default sic@ */

    /* @sic R5s150180: tsc_IMS_VideoFeature replaced by pc_IMS_Video_FeatureTag sic@ */

    /* @sic R5-160793: p_AdditionalOptionTagsForSupported removed as Supported header contains "precondion" in all cases where this function is used sic@ */

    /* @sic R5-174544: SessionId removed sic@ */
    /* @sic R5s170XXX: p_AdditionalOptionTagsForSupported restored as "precondion" is not included in C.11c sic@ */
    var template (value) MessageHeader v_MessageHeader_Invite := cs_MessageHeader_Dummy;

    var template (value) SipUrl v_PublicUserId_SipUrl := f_IMS_PTC_ImsInfo_GetSipUriTX(defaultId);  // @sic R5-133304: use default public user id sic@

    var template (omit) SemicolonParam_List v_ContactParams := omit;

    var template (value) GenericParam v_GenericParam_3gpp_icsi_ref;

    var charstring v_CallIdString := f_IMS_PTC_GenerateCallId("INVITE");

    var charstring v_FromTag := f_IMS_GenerateTag("INVITE", "From");

    var template (value) RouteBody_List v_RouteBodyList;

    var integer v_CSeqValue;

    var template (value) From v_From;

    var template (value) To v_To;

    var template (value) CallId v_CallId;

    select (p_Context) {

      case (A_2_1_A4) {

        v_CSeqValue := p_CSeqValue;

        v_From   := cs_From(f_SIP_BuildSipUri_TX(tsc_IMS_CallerSipAddrStr), v_FromTag);

        v_To     := cs_ToDef(v_PublicUserId_SipUrl);

        v_CallId := cs_CallId(v_CallIdString);

        v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();      /* @sic R5s140324 - MCC160 Implementation sic@ */

      }

      case (A_2_1_A5) {

        v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogLocal);

        v_From   := f_IMS_RequestInDialog_FromHeaderTX();

        v_To     := f_IMS_RequestInDialog_ToHeaderTX();

        v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

        if (f_IMS_PTC_ImsInfo_DialogIsMO()) {      /* @sic R5s140324 - MCC160 Implementation sic@ */

          v_RouteBodyList := fl_RouteSet_ReverseOrder(f_IMS_RouteSet_MO_Call_TX());    /* @sic R5-142320, R5-144601 sic@ */

        } else {

          v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "unknown context");

      }

    }

    v_MessageHeader_Invite.via             := f_IMS_InviteRequest_ViaTX();  /* @sic R5-134116: New function fl_IMS_InviteRequest_ViaTX sic@ */

    v_MessageHeader_Invite.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;

    v_MessageHeader_Invite.toField         := v_To;

    v_MessageHeader_Invite.callId          := v_CallId;

    v_MessageHeader_Invite.cSeq            := cs_CseqDef(v_CSeqValue, "INVITE");



    v_MessageHeader_Invite.supported       := cs_Supported(tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported);
    v_MessageHeader_Invite.pCalledPartyID  := cs_PCalledPartyId(v_PublicUserId_SipUrl);

    v_MessageHeader_Invite.contentType     := cs_ContentTypeSDP;

    v_MessageHeader_Invite.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Invite.accept          := cs_Accept({cs_AcceptBody(c_sdpApplication), cs_AcceptBody(c_ims3gppApplication)});

    if (p_MTSI) {

      v_GenericParam_3gpp_icsi_ref := cs_GenericParam_Common(tsc_IMS_FeatureParamId_3gpp_icsi_ref, cs_GenValueQuoted(cs_IMS_EncodedFeatureParamValue_3gpp_icsi_ref));  // @sic R5s130738: string needs to be quoted; R5-142247 sic@

      if (p_Context == A_2_1_A4) {      // @sic R5-142949 sic@

        v_MessageHeader_Invite.pAssertedService := cs_PAssertedService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 1 sic@ */

        v_MessageHeader_Invite.acceptContact    := cs_AcceptContact({v_GenericParam_3gpp_icsi_ref});

      }

      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, v_GenericParam_3gpp_icsi_ref);

    }

    if (p_Video) { // @sic R5s130109 sic@

      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));

    }

    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                     /* @sic R5-165943 sic@ */

    v_MessageHeader_Invite.contact := cs_Contact(f_SIP_BuildSipUri_TX(tsc_IMS_CallerSipAddrStr, tsc_IMS_CallerPort), v_ContactParams);

    if (p_Context != A_2_1_A5) {

      if (p_SrvccAlertingSupported) {   /* @sic R5s140385 change 6 - NOTE: feature caps shall only be inserted when creating a voice call sic@ */

        v_MessageHeader_Invite.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccAlerting)});  // @sic R5-145796 sic@

      }

      f_IMS_PTC_ImsInfo_DialogInit(dialogLocal, valueof(v_MessageHeader_Invite));

    }

    return v_MessageHeader_Invite;

  }

	  function f_IMS_MTCallSetup_AnnexC11_INVITE(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,

                                             boolean p_PreconditionsRequired := true) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5w140112: return IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ; initialisation of routing info sic@ */

    /* @sic R5-153798: p_AdditionalAudioCodecs sic@ */

    /* @sic R5-169060: allow no preconditions sic@ */

    /* @sic R5-173535: p_AudioCodecList instead of p_AdditionalAudioCodecs sic@ */

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var template (value) INVITE_Request v_InviteRequest;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) SDP_Message v_SDP_MessageStep1;
    var OptionTag_List v_AdditionalOptionTagsForSupported;
    if (p_PreconditionsRequiredAndSupported) {

      v_AdditionalOptionTagsForSupported := tsc_OptionTagList_precondition;

    } else {

      v_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty;

    }
    v_SDP_MessageStep1 := f_IMS_BuildSDP_MTSpeechCall_Step1(p_AudioCodecList, p_PreconditionsRequired);

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX(v_AdditionalOptionTagsForSupported);

    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(v_SDP_MessageStep1));

    return f_IMS_InviteRequestWithSdpMT(v_InviteRequest, v_SDP_MessageStep1);

  }


Change 3 – Enhancement of f_IMS_BuildSDP_MTSpeechCall_Step4or8 
In order to re-use common TTCN code for FBBA f_IMS_BuildSDP_MTSpeechCall_Step4or8 has been enhanced to support SDP message without preconditions
	  function f_IMS_BuildSDP_MTSpeechCall_Step4or8(charstring p_Step,

                                                IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                                boolean p_PreconditionsRequired := true)

    runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5-123077 change 3: 'inactive' removed from 'Attributes for media' for step 4 sic@ */

    /* @sic R5s150704, R5s150751 - Additional changes: f_IMS_BuildSDP_RX used instead of cr_SDP_Message_Def sic@ */

    /* @sic R5-153798: p_IsC45 sic@ */

    /* @sic R5-168057: new default for p_AudioCodec sic@ */

    var template (present) SDP_media_field v_SDP_Media_Field := cr_SDP_Media_Audio;

    var template (present) SDP_attribute_rtpmap_codec v_RtpmapCodec := f_SDP_AttributeRtpmapVoiceRX(p_AudioCodec);   /* @sic R5-153798 sic@*/

    var template (present) SDP_attribute_fmtp_format_specific_params v_FormatSpecificParameters := ?;

    var template (present) SemicolonParam_List v_ParamList;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes;

    select (p_Step) {

      case ("step 4") {

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes((c_none, c_sendrecv), c_none, c_mandatory, c_mandatory);     // @sic R5s120530 change 3.15 sic@

        v_SDP_PrecondionAttributes := f_SDP_Attributes_Add_RX(v_SDP_PrecondionAttributes, cr_SDP_Attribute_conf_qos);          // @sic R5s140204 sic@

        if (p_AudioCodec == EVS_16000) {    // C.45 @sic R5-153798 sic@

          v_ParamList := {

            cr_GenericParam("br-send", ?),

            cr_GenericParam("br-recv", ?),

            cr_GenericParam("bw-send", ?),  /* @sic R5-174050: any value for bw-send sic@ */

            cr_GenericParam("bw-recv", ?),  /* @sic R5-174050: any value for bw-recv sic@ */

            *

          };

          v_FormatSpecificParameters := cr_Fmtp_ParamList(v_ParamList);

        }

      }

      case ("step 8") {

        /* @sic R5-150709: cr_SDP_Attribute_sendrecv removed sic@ */

        v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    v_SDP_MediaAttributes := {

      cr_SDP_Attribute_rtpmap(cr_SDP_AnyMediaFormatDescr, v_RtpmapCodec),    // @sic R5s130681 change 2 sic@

      cr_SDP_Attribute_fmtp(cr_SDP_AnyMediaFormatDescr, v_FormatSpecificParameters)

    };

    if (not p_PreconditionsRequired) {

      v_SDP_PrecondionAttributes := {};
    }
    return f_IMS_BuildSDP_RX(-, v_SDP_Media_Field, v_SDP_MediaAttributes, v_SDP_PrecondionAttributes);

  }


Change 4 – Common function for Call-Hold test cases
In order to have a common test case body function for Call-Hold test cases in EUTRA and WLAN the respective radio bearer specific functions are made generic and test case implementation of TC_15_11, TC_15_11a, TC_15_12, TC_15_12a and the corresponding WLAN test cases is changed accordingly:

	  function f_IMS_MO_CallOnHoldWithoutAnnouncement(INVITE_Request p_InviteRequest,

                                                  MessageHeader p_MessageHeader_Response183,

                                                  SDP_Message p_SDP_OriginalOffer) runs on IMS_PTC

  {





    var SDP_Message v_SDP_HoldResumeOffer;








    f_IMS_TestBody_Set(true);

    // @siclog "Step 1 - 4" siclog@

    f_UT_CallHold(MMI);

    v_SDP_HoldResumeOffer := f_IMS_MOCallHoldResume_AnnexC8(HOLD, p_InviteRequest, p_MessageHeader_Response183, p_SDP_OriginalOffer);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 5 - 8" siclog@

    f_UT_CallResume(MMI);

    f_IMS_MOCallHoldResume_AnnexC8(RESUME, p_InviteRequest, p_MessageHeader_Response183, p_SDP_OriginalOffer);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 9 -10" siclog@

    f_IMS_CallReleaseMO(p_InviteRequest);

    f_IMS_TestBody_Set(false);

  }






























	  function f_IMS_MT_CallOnHoldWithoutAnnouncement(INVITE_Request p_InviteRequest,

                                                  SipUrl p_ContactUrl,

                                                  template (value) SDP_Message p_SDP_LatestMessageBySS) runs on IMS_PTC

  {





    var template (value)  SDP_Message v_SDP_LatestMessageBySS;











    f_IMS_TestBody_Set(true);

    // @siclog "Step 1 - 4" siclog@

    fl_IMS_MTCallHoldResume_AnnexC9(HOLD, p_ContactUrl, p_SDP_LatestMessageBySS);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 5 - 8" siclog@

    fl_IMS_MTCallHoldResume_AnnexC9(RESUME, p_ContactUrl, p_SDP_LatestMessageBySS);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 9 -10" siclog@

    f_IMS_CallReleaseMT(p_InviteRequest, p_ContactUrl);

    f_IMS_TestBody_Set(false);

  }
































	  function f_TC_15_11_IMS1() runs on IMS_PTC

  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_Response183;

    var SDP_Message v_SDP_OriginalOffer;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

     // setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);  /* @sic R5w140112 sic@ */

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    v_SDP_OriginalOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183);   // v_SDP_OriginalOffer: SDP offer sent by the UE in step 5 or 7 of C.21

    f_IMS_MO_CallOnHoldWithoutAnnouncement(v_InviteRequest, v_MessageHeader_Response183, v_SDP_OriginalOffer);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }

	  function f_TC_15_11a_IMS1() runs on IMS_PTC

  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_Response183;

    var SDP_Message v_SDP_Offer;
    f_IMS_CC_Preamble(IPCAN_VideoCall, IMS_REGISTERED);
    // setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_VideoCall);

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(SetupVideoCall);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_SDP_Offer := v_InviteRequestWithSdp.SdpOffer;

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp);

    // See TS. 36.508 EUTRA/EPS signalling for IMS MO video call Table 4.5A.8.3-1 Step 12-15

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_VideoCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearers

    v_SDP_Offer := f_IMS_MOCallSetup_AnnexC25_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_MO_CallOnHoldWithoutAnnouncement(v_InviteRequest, v_MessageHeader_Response183, v_SDP_Offer);
    f_IMS_CC_Postamble(IPCAN_MO_VideoCall);

  }

	  function f_TC_15_12_IMS1() runs on IMS_PTC

  {

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    // setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);                                                  // C.21 Step 1

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();                          // C.21 Step 2

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_ContactUrl := f_Addr_Union_GetSipUrl(v_InviteRequest.msgHeader.contact.contactBody.contactAddresses[0].addressField);

    v_MessageHeader_183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);        // C.21 Step 3, 4

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    v_SDP_LatestMessageByUE := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_183); // @sic R5s120609 sic@

    v_SDP_LatestMessageBySS := f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestMessageByUE);   // as in f_IMS_MOCallSetup_AnnexC21_Steps5_13

    f_IMS_MT_CallOnHoldWithoutAnnouncement(v_InviteRequest, v_ContactUrl, v_SDP_LatestMessageBySS);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }

	  function f_TC_15_12a_IMS1() runs on IMS_PTC

  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_Response183;

    var SDP_Message v_SDP_UE;

    var template (value) SDP_Message v_SDP_SS;

    var SipUrl v_ContactUrl;
    f_IMS_CC_Preamble(IPCAN_VideoCall, IMS_REGISTERED);
    // setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_VideoCall);  // C.25 Step 1

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(SetupVideoCall);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_ContactUrl := f_Addr_Union_GetSipUrl(v_InviteRequest.msgHeader.contact.contactBody.contactAddresses[0].addressField);

    v_SDP_UE := v_InviteRequestWithSdp.SdpOffer;

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp);

    // See TS. 36.508 EUTRA/EPS signalling for IMS MO video call Table 4.5A.8.3-1 Step 12-15

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_VideoCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearers

    v_SDP_UE := f_IMS_MOCallSetup_AnnexC25_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    v_SDP_SS := f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_UE);
    f_IMS_MT_CallOnHoldWithoutAnnouncement(v_InviteRequest, v_ContactUrl, v_SDP_SS);

    f_IMS_CC_Postamble(IPCAN_MO_VideoCall);

  }


Change 5 – Enhancement of f_IMS_CC_MO_SpeechCall
Several test cases need an MO call to be established as initial conditions
( f_IMS_CC_MO_SpeechCall has been enhanced to support other IPCANs

	  function f_IMS_MOCallSetup_AnnexC21C21b(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,

                                          IMS_SDP_AudioCodec_Type p_AudioCodecToBeSelected := AMR_16000,

                                          integer p_BandwidthAS := 37) runs on IMS_PTC return INVITE_Request

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;
    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();
    // Step 2
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4(-, -, p_AudioCodecList);

    // Step 3, 4

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, -, -, -, p_AudioCodecToBeSelected, p_BandwidthAS);  /* @sic R5w140112 sic@ */

    // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183, -, -, -, p_AudioCodecToBeSelected);

    return v_InviteRequestWithSdp.Invite;

  }

	  function f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b(IMS_TestProcedure_Type p_TestProcedure,

                                                    IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,

                                                    boolean p_PreconditionsRequired := true) runs on IMS_PTC return INVITE_Request

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected := p_AudioCodecList[0];

    // @siclog "Step 2" siclog@

    select (p_TestProcedure) {

      case (IPCAN_MO_SpeechCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_Common(-, -, -, p_AudioCodecList, -, p_PreconditionsRequired);

      }

      case (IPCAN_MO_VideoCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(SetupVideoCall);

      }

    }

    // @siclog "Step 3" siclog@

    f_IMS_MOCallSetup_AnnexC21_Step3(v_InviteRequestWithSdp);

    // @siclog "Step 4 - 6" siclog@

    f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b_Steps4_6(v_InviteRequestWithSdp, -, p_PreconditionsRequired, v_AudioCodecToBeSelected);

    // @siclog "Step 7 - 8" siclog@

    f_IMS_MOCallSetup_AnnexC21C25_Step12_13(v_InviteRequestWithSdp);

    return v_InviteRequestWithSdp.Invite;

  }

	  function f_IMS_CC_MO_SpeechCall(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,

                                  IMS_SDP_AudioCodec_Type p_AudioCodecToBeSelected := AMR_16000,

                                  boolean p_PreconditionsRequired := true) runs on IMS_PTC return INVITE_Request

  {

    var INVITE_Request v_InviteRequest;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    if (not p_PreconditionsRequired) {

      FatalError(__FILE__, __LINE__, "not supported");

    }
    // Step 1

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    if (v_IsWLAN) {

      v_InviteRequest := f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b(IPCAN_MO_SpeechCall, p_AudioCodecList);

    }

    else {

      v_InviteRequest := f_IMS_MOCallSetup_AnnexC21C21b(p_AudioCodecList, p_AudioCodecToBeSelected);

    }

    return v_InviteRequest;


  }


Change 6 – Enhancement of f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9
In order to support WLAN conferencing test cases conference call creation need to consider call establishment procedures for WLAN.

NOTE: A prose CR is needed as in current prose WLAN test cases G.15.18 and G.15.21 refer to test cases 15.19 and 15.19a, which refer to C.10 and C.38, which refer to C.21 and C.25 – but for WLAN references to C.21a and C.25a is required.
	  function f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9(IMS_TestProcedure_Type p_TestProcedure,

                                                         charstring p_ConferenceFactoryUri,

                                                         charstring p_TemporaryConferenceUri,

                                                         charstring p_FinalConferenceUri,

                                                         boolean p_RadioBearersEstablished := false) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var template (omit) SemicolonParam_List v_AdditionalContactParams := {cs_GenericParam("isfocus")}; // As per C.10

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_183;

    var template (value) SipUrl v_ContactUrl;


    var template (value) MessageHeader v_MessageHeader_InviteResponse_200OK;

    var IMS_PreconditionIndication_Type v_PreconditionIndication_200OK;

    var template (value) SDP_Message v_SDP_Message_200OK;

    var template (omit)  MessageBody v_MessageBody_200OK;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (present) SipUrl v_ConferenceFactorySipUrl := f_SIP_BuildSipUri_RX(p_ConferenceFactoryUri);

    var IMS_CallType_Type v_CallType := CONFERENCE_CALL;  /* @sic R5-155179 sic@ */

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();
    // @siclog "Step 2" siclog@         UE sends INVITE

    select (p_TestProcedure) {     // @sic R5s150373 change 1 sic@

      case (IPCAN_MO_SpeechCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_Common(v_ConferenceFactorySipUrl);

      }

      case (IPCAN_MO_VideoCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(SetupVideoCall, v_ConferenceFactorySipUrl);

      }

    }

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsWLAN) {

      f_IMS_MOCallSetup_AnnexC21_Step3(v_InviteRequestWithSdp);

      if (not p_RadioBearersEstablished) {

        f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);

      }

      select (p_TestProcedure) {

        case (IPCAN_MO_SpeechCall) {

          v_SDP_Message_200OK := f_IMS_BuildSDP_AnnexC21_Step4(v_InviteRequestWithSdp.SdpOffer);

        }

        case (IPCAN_MO_VideoCall) {

          v_SDP_Message_200OK := f_IMS_BuildSDP_MOCallAudioVideo_Step4(SetupVideoCall, v_InviteRequestWithSdp.SdpOffer);

        }

      }

      v_MessageBody_200OK := cs_MessageBody_SDP(v_SDP_Message_200OK);

      v_PreconditionIndication_200OK := PreconditionRequired;

    } 

    else {
    // @siclog "Step 3 - 4" siclog@

    select (p_TestProcedure) {

      case (IPCAN_MO_SpeechCall) {  // same steps as for C.10 what in fact are steps 3 - 4 of C.21

        v_MessageHeader_183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, p_TemporaryConferenceUri, v_AdditionalContactParams, v_CallType);

      }

      case (IPCAN_MO_VideoCall) {   // same steps as for C.10 what in fact are steps 3 - 4 of C.25

        v_MessageHeader_183 := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp, p_TemporaryConferenceUri, v_AdditionalContactParams, v_CallType);

      }

    }

    if (not p_RadioBearersEstablished) {

      // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

      f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    }

    // @siclog "Step 5 - 7a" siclog@

    select (p_TestProcedure) {

      case (IPCAN_MO_SpeechCall) {    // same steps as steps 5 - 8 of C.21

        f_IMS_MOCallSetup_AnnexC21_Steps5_8(v_InviteRequestWithSdp, v_MessageHeader_183, -, -, v_CallType);

      }

      case (IPCAN_MO_VideoCall) {     // same steps as steps 5 - 8 of C.25

        f_IMS_MOCallSetup_Video_Steps5_8(SetupVideoCall, v_InviteRequestWithSdp, v_MessageHeader_183, v_CallType);        // @sic R5s150373 change 1: C25 instead of AddVideo sic@

      }

    }
      v_MessageBody_200OK := omit;

      v_PreconditionIndication_200OK := NoPreconditionIndication;

    }
    // @siclog "Step 8" siclog@         Send 200 OK (response to INVITE)


    v_MessageHeader_InviteResponse_200OK := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, -, -, v_PreconditionIndication_200OK, v_CallType);
    v_ContactUrl := f_SIP_BuildSipUri_TX(p_FinalConferenceUri);

    v_MessageHeader_InviteResponse_200OK.contact := cs_Contact(v_ContactUrl, v_AdditionalContactParams);   /* @sic R5-151681 sic@ */

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);


    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200OK, v_MessageBody_200OK)));
    // @siclog "Step 9" siclog@         UE sends an ACK to acknowledge receipt of the 200 OK for INVITE

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_CallType);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    return v_InviteRequestWithSdp;

  }


Change 7 – Enhancement of f_IMS_MOCallwithOIR
f_IMS_MOCallwithOIR has not been implemented correctly according to the prose for WLAN and FBBA.

NOTE: This change should not affect the implementation for EUTRA.
	  function f_IMS_MOCallwithOIR(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var MessageHeader v_MessageHeader_Invite;

    var MessageHeader v_MessageHeader_Response183;

    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected := p_AudioCodecList[0];

    var boolean v_OriginatingIdentificationRestriction := true;

    var boolean v_IsC21aC21c := f_IMS_PTC_RanTypeIsWLAN();


    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, -, normalCallOIR);

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_Common(-, -, -, p_AudioCodecList, v_OriginatingIdentificationRestriction);

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    fl_MessageHeader_CheckFromAnonymous(v_MessageHeader_Invite);

    fl_MessageHeader_CheckPrivacy(v_MessageHeader_Invite);





    if (v_IsC21aC21c) {

      // @siclog "Step 3" siclog@

      f_IMS_MOCallSetup_AnnexC21_Step3(v_InviteRequestWithSdp);

      // @siclog "Step 4 - 6" siclog@ Annex C21a/C21b step 4-6.

      f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b_Steps4_6(IPCAN_MO_SpeechCall, v_InviteRequestWithSdp, -, -, v_AudioCodecToBeSelected); 

      // @siclog "Step 7 - 8" siclog@ Annex C21a/C21b step 7-8.

      f_IMS_MOCallSetup_AnnexC21C25_Step12_13(v_InviteRequestWithSdp);

    } 

    else {
      // @siclog "Step 3 - 4" siclog@ Annex C21

      v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, -, -, -, v_AudioCodecToBeSelected);





      // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

      f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer








      // @siclog "Step 5 - 13" siclog@ Annex C21

      f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183, -, -, -, v_AudioCodecToBeSelected);

    }

    // Steps defined in Annex C.32

    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);

    f_IMS_TestBody_Set(false);

  }


Change 8 – Enhancement of f_IMS_MTCallwithTIR
f_IMS_MTCallwithTIR has not been implemented correctly according to the prose for WLAN and FBBA.

NOTE: This change should not affect the implementation for EUTRA.
	  function f_IMS_MTCallwithTIR(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var template (omit) IMS_ResponseWithSdp_Type v_OptionalResponseWithSdp;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionProgress183;

    var template (present) SDP_Message v_SDP_Message;
    var MessageHeader v_MessageHeader_SessionProgress;

    var template (omit) MessageHeader v_MessageHeader_Ringing;

    var MessageHeader v_MessageHeader_200OK;

    var SipUrl v_ContactUrl;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();
    var boolean v_IsC11aC11c := v_IsFBBA or v_IsWLAN;
    var boolean v_PreconditionsRequired := not v_IsFBBA;

    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected := p_AudioCodecList[0];

    timer t_Dummy;
    f_IMS_TestBody_Set(true);

    f_UT_MMI_ActivateIMS_TIR(MMI);

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE(p_AudioCodecList, v_PreconditionsRequired);

    if (v_IsC11aC11c) {

      // @siclog "Step 1" siclog@

      IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequestWithSdp.Invite));

      // @siclog "Step 2 - 5" siclog@

      v_SDP_Message := f_IMS_BuildSDP_C11aC11c_Step3(v_AudioCodecToBeSelected, v_PreconditionsRequired);      

      v_OptionalResponseWithSdp := f_IMS_MTCallSetup_AnnexC11aC11cC26aC26b_Steps2_5(t_Dummy, v_InviteRequestWithSdp, v_SDP_Message, v_InviteRequestWithSdp.RoutingInfo.Protocol);

      f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  /* trigger IPCAN to do the routing for RTP/RTCP loopback */

      if (ispresent(v_OptionalResponseWithSdp)) {

        v_ResponseWithSdp := valueof(v_OptionalResponseWithSdp);

        v_MessageHeader_Ringing := v_ResponseWithSdp.Response.msgHeader;

      }

    } else {
    // @siclog "Step 1 - 4" siclog@


    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp, v_AudioCodecToBeSelected, v_PreconditionsRequired);

    v_SessionProgress183 := v_ResponseWithSdp.Response;

    v_MessageHeader_SessionProgress := v_SessionProgress183.msgHeader;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionProgress183.msgHeader);

    fl_MessageHeader_CheckPrivacy(v_MessageHeader_SessionProgress);




      // Establish Dedicated EPS bearer

      // See TS. 36.508 EUTRA/EPS signalling for IMS MT speech call Table 4.5A.7.3-1 Step 9-12

      f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);


    // Step 5 - 6. Send PRACK, Receive 200 OK

    f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequestWithSdp.Invite, v_SessionProgress183.msgHeader);

    // Step 7 - 11. Send UPDATE, Receive 200 OK, Receive optional 180 Ringing and send PRACK if requested

    v_MessageHeader_Ringing := f_IMS_MTCallSetup_AnnexC11_Steps7_11(v_InviteRequestWithSdp, v_ResponseWithSdp, v_AudioCodecToBeSelected);

    if (ispresent(v_MessageHeader_Ringing)) {

      fl_MessageHeader_CheckPrivacy(valueof(v_MessageHeader_Ringing));

    }

    // Step 11A - Make UE accept the speech AMR offer

    f_UT_AnswerCall(MMI);

    // Step 12. Receive 200 OK, Send ACK

    v_ResponseWithSdp := f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequestWithSdp);

    v_MessageHeader_200OK := v_ResponseWithSdp.Response.msgHeader;

    fl_MessageHeader_CheckPrivacy(v_MessageHeader_200OK);

    // MT Call is now established

    f_Delay(3.0);    /* @sic R5s130510 change 1; draft prose CR sic@ */

    // Step 14-15. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    // @siclog "Step 14 -15" siclog@

    f_IMS_CallReleaseMT(v_InviteRequestWithSdp.Invite, v_ContactUrl);

    f_UT_MMI_DeactivateIMS_TIR(MMI);

    f_IMS_TestBody_Set(false);

  }


Change 9 – Common function for three way session test cases

In order to have a common test case body function for three way session test cases in EUTRA and WLAN the common test body function f_TC_15_21a_TC_15_21c_Body has been enhanced to support WLAN as well.

a) New function fl_IMS_MOCallSetup_Step2 to be used for session establishment in conferencing test cases
	  function fl_IMS_MOCallSetup_Step2(IMS_TestProcedure_Type p_TestProcedure,

                                    charstring p_CalleeUri) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var template (present) SipUrl v_CalleeSipUrl := f_SIP_BuildSipUri_RX(p_CalleeUri);

    select (p_TestProcedure) {

      case (IPCAN_MO_SpeechCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_Common(v_CalleeSipUrl);

      }

      case (IPCAN_MO_VideoCall) {

        v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(SetupVideoCall, v_CalleeSipUrl);

      }

    }

    return v_InviteRequestWithSdp;

  }

	  function f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9(IMS_TestProcedure_Type p_TestProcedure,

                                                         charstring p_ConferenceFactoryUri,

                                                         charstring p_TemporaryConferenceUri,

                                                         charstring p_FinalConferenceUri,

                                                         boolean p_RadioBearersEstablished := false) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var template (omit) SemicolonParam_List v_AdditionalContactParams := {cs_GenericParam("isfocus")}; // As per C.10

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_183;

    var template (value) SipUrl v_ContactUrl;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200OK;

    var IMS_PreconditionIndication_Type v_PreconditionIndication_200OK;

    var template (value) SDP_Message v_SDP_Message_200OK;

    var template (omit)  MessageBody v_MessageBody_200OK;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Ack;


    var IMS_CallType_Type v_CallType := CONFERENCE_CALL;  /* @sic R5-155179 sic@ */

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();

    // @siclog "Step 2" siclog@         UE sends INVITE

    v_InviteRequestWithSdp := fl_IMS_MOCallSetup_Step2(p_TestProcedure, p_ConferenceFactoryUri);








    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsWLAN) {

      f_IMS_MOCallSetup_AnnexC21_Step3(v_InviteRequestWithSdp);

      if (not p_RadioBearersEstablished) {

        f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);

      }

      select (p_TestProcedure) {

        case (IPCAN_MO_SpeechCall) {

          v_SDP_Message_200OK := f_IMS_BuildSDP_AnnexC21_Step4(v_InviteRequestWithSdp.SdpOffer);

        }

        case (IPCAN_MO_VideoCall) {

          v_SDP_Message_200OK := f_IMS_BuildSDP_MOCallAudioVideo_Step4(SetupVideoCall, v_InviteRequestWithSdp.SdpOffer);

        }

      }

      v_MessageBody_200OK := cs_MessageBody_SDP(v_SDP_Message_200OK);

      v_PreconditionIndication_200OK := PreconditionRequired;

    } 

    else {

      // @siclog "Step 3 - 4" siclog@

      select (p_TestProcedure) {

        case (IPCAN_MO_SpeechCall) {  // same steps as for C.10 what in fact are steps 3 - 4 of C.21

          v_MessageHeader_183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, p_TemporaryConferenceUri, v_AdditionalContactParams, v_CallType);

        }

        case (IPCAN_MO_VideoCall) {   // same steps as for C.10 what in fact are steps 3 - 4 of C.25

          v_MessageHeader_183 := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp, p_TemporaryConferenceUri, v_AdditionalContactParams, v_CallType);

        }

      }

      if (not p_RadioBearersEstablished) {

        // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

        f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

      }

      // @siclog "Step 5 - 7a" siclog@

      select (p_TestProcedure) {

        case (IPCAN_MO_SpeechCall) {    // same steps as steps 5 - 8 of C.21

          f_IMS_MOCallSetup_AnnexC21_Steps5_8(v_InviteRequestWithSdp, v_MessageHeader_183, -, -, v_CallType);

        }

        case (IPCAN_MO_VideoCall) {     // same steps as steps 5 - 8 of C.25

          f_IMS_MOCallSetup_Video_Steps5_8(SetupVideoCall, v_InviteRequestWithSdp, v_MessageHeader_183, v_CallType);        // @sic R5s150373 change 1: C25 instead of AddVideo sic@

        }

      }

      v_MessageBody_200OK := omit;

      v_PreconditionIndication_200OK := NoPreconditionIndication;

    }

    // @siclog "Step 8" siclog@         Send 200 OK (response to INVITE)

    v_MessageHeader_InviteResponse_200OK := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, -, -, v_PreconditionIndication_200OK, v_CallType);

    v_ContactUrl := f_SIP_BuildSipUri_TX(p_FinalConferenceUri);

    v_MessageHeader_InviteResponse_200OK.contact := cs_Contact(v_ContactUrl, v_AdditionalContactParams);   /* @sic R5-151681 sic@ */

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200OK, v_MessageBody_200OK)));

    // @siclog "Step 9" siclog@         UE sends an ACK to acknowledge receipt of the 200 OK for INVITE

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_CallType);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    return v_InviteRequestWithSdp;

  }


b) Common implementation for three way session creation
	  type record ThreeWaySession_SessionInfo_Type {

    IMS_InviteRequestWithSdp_Type InviteRequestWithSdp,     // INVITE sent by the UE to create the session

    MessageHeader MessageHeader_PrevResponse,               // earlier response sent to the UE within the same dialog

    SDP_Message LatestSdpOffer,                             // latest SDP offer sent by the UE during session establishment

    SipUrl ContactUrl

  };

	  function fl_IMS_ThreeWaySession_CreateSession(IMS_TestProcedure_Type p_TestProcedure,

                                                charstring p_CalleeUri,

                                                charstring p_CalleeContactUri,

                                                boolean p_RadioBearersEstablished) runs on IMS_PTC return ThreeWaySession_SessionInfo_Type
  {

    var ThreeWaySession_SessionInfo_Type v_SessionInfo;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();                                                                              

    v_SessionInfo.InviteRequestWithSdp := fl_IMS_MOCallSetup_Step2(p_TestProcedure, p_CalleeUri);     // C.21/25/21a/25a Step 2

    if (v_IsWLAN) {

      v_ResponseWithSdp := f_IMS_MOCallSetup_AnnexC21aC21cC25aC25b_Steps3_8(p_TestProcedure, v_SessionInfo.InviteRequestWithSdp, -, -, p_RadioBearersEstablished);

      v_SessionInfo.MessageHeader_PrevResponse := v_ResponseWithSdp.Response.msgHeader;

      v_SessionInfo.LatestSdpOffer := v_SessionInfo.InviteRequestWithSdp.SdpOffer

    } 

    else {

      select (p_TestProcedure) {

        case (IPCAN_MO_SpeechCall) {

          v_SessionInfo.MessageHeader_PrevResponse := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_SessionInfo.InviteRequestWithSdp, p_CalleeContactUri);                                   // C.21 Step 3, 4

          if (not p_RadioBearersEstablished) {

            f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

          }

          v_SessionInfo.LatestSdpOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_SessionInfo.InviteRequestWithSdp, v_SessionInfo.MessageHeader_PrevResponse, p_CalleeContactUri);   // C.21 Step 5 .. 13

        }

        case (IPCAN_MO_VideoCall) {

          v_SessionInfo.MessageHeader_PrevResponse := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_SessionInfo.InviteRequestWithSdp, p_CalleeContactUri);                      // C.25 Step 3, 4

          if (not p_RadioBearersEstablished) {

            f_IMS_CC_TriggerDedicatedBearerActivation(p_TestProcedure);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

          }

          v_SessionInfo.LatestSdpOffer := f_IMS_MOCallSetup_AnnexC25_Steps5_13(v_SessionInfo.InviteRequestWithSdp, v_SessionInfo.MessageHeader_PrevResponse, p_CalleeContactUri);   // C.25 Step 5 .. 13

        }

      }

    }

    v_SessionInfo.ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInfo.MessageHeader_PrevResponse);

    return v_SessionInfo;

  }

	  function f_IMS_ThreeWaySession_Creation(IMS_TestProcedure_Type p_TestProcedure) runs on IMS_PTC

  {

    var charstring v_ConferenceFactoryUri   := fl_IMS_BuildConferenceUri(conferenceFactoryUri);             // @sic R5-150745 sic@

    var charstring v_TemporaryConferenceUri := fl_IMS_BuildConferenceUri(temporaryConferenceUri);           // @sic R5-150745 sic@

    var charstring v_FinalConferenceUri     := fl_IMS_BuildConferenceUri(finalConferenceUri);               // @sic R5-150745 sic@

    var charstring v_CalleeUri_Dialog1 := px_IMS_CalleeUri;

    var charstring v_CalleeUri_Dialog2 := px_IMS_CalleeUri2;                            // @sic R5-151710 sic@

    var charstring v_CalleeContactUri_Dialog1 := px_IMS_CalleeContactUri;

    var charstring v_CalleeContactUri_Dialog2 := px_IMS_CalleeContactUri2;              // @sic R5-151710 sic@





    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog1;

    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog2;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp_Dialog3;



    var ConferenceEventPackage_SubscribeInfo_Type v_ConferenceEventPackage_SubscribeInfo_ByRef := valueof(cs_ConferenceEventPackage_SubscribeInfo_Init);

    var AnnexC19C37_StateInfo_Type v_AnnexC19C37_Dialog1_ByRef := valueof(cs_AnnexC19C37_StateInfo_Init(v_CalleeUri_Dialog1, v_CalleeContactUri_Dialog1));

    var AnnexC19C37_StateInfo_Type v_AnnexC19C37_Dialog2_ByRef := valueof(cs_AnnexC19C37_StateInfo_Init(v_CalleeUri_Dialog2, v_CalleeContactUri_Dialog2));



    var SipUrl v_ContactUrl_UE;

    var template (omit) SUBSCRIBE_Request v_SubscribeRequest := omit;

    var template (present) MessageHeader v_MessageHeader_Bye_Dialog1;

    var template (present) MessageHeader v_MessageHeader_Bye_Dialog2;

    var SDP_Message v_SDP_OriginalOffer_Dialog1;

    var SDP_Message v_SDP_OriginalOffer_Dialog2;

    var boolean v_RadioBearersEstablished := false;

    var boolean v_WaitForBYE_Dialog1 := false;

    var boolean v_WaitForBYE_Dialog2 := false;

    var boolean v_Dialog1_Terminated := false;

    var boolean v_Dialog2_Terminated := false;

    var boolean v_Continue;

    var integer v_ConferenceEventPackage_Version := 1;   // version used in notifications sent to the UE to inform about remote user has joined the conference

    //--------------------------------------------

    // DIALOG 1

    //--------------------------------------------

    // Setup MO speech call

    f_IMS_CC_StartCall(p_TestProcedure, v_CalleeUri_Dialog1);                                                   // C.21/25 Step 1

    v_SessionInfo_Dialog1 := fl_IMS_ThreeWaySession_CreateSession(p_TestProcedure, v_CalleeUri_Dialog1, v_CalleeContactUri_Dialog1, v_RadioBearersEstablished);















    v_RadioBearersEstablished := true;

    f_IMS_TestBody_Set(true);

    //Steps 1 to 4 - Messages in Annex C.8 - Set ongoing call on hold

    // @siclog "Step 1 - 4" siclog@

    f_UT_CallHold(MMI);

    f_IMS_MOCallHoldResume_AnnexC8(HOLD, 
                                         v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog1.MessageHeader_PrevResponse, v_SessionInfo_Dialog1.LatestSdpOffer, v_CalleeContactUri_Dialog1);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    //--------------------------------------------

    // DIALOG 2

    //--------------------------------------------

    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);

    // @siclog "Step 5 - 17" siclog@    Steps defined in Annex C.21 - MO call
    v_SessionInfo_Dialog2 := fl_IMS_ThreeWaySession_CreateSession(p_TestProcedure, v_CalleeUri_Dialog2, v_CalleeContactUri_Dialog2, v_RadioBearersEstablished);

















    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 17A" siclog@

    f_UT_CallMPTY(MMI);  // There is no IPCAN signalling involved => f_IMS_CC_StartCall shall not be used

    // @siclog "Step 17B - 17E" siclog@

    f_IMS_MOCallHoldResume_AnnexC8(HOLD, 
                                         v_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog2.MessageHeader_PrevResponse, v_SessionInfo_Dialog2.LatestSdpOffer, v_CalleeContactUri_Dialog2);
    //--------------------------------------------

    // DIALOG 3

    //--------------------------------------------

    f_IMS_PTC_ImsInfo_ActivateDialog(thirdDialog);

    // @siclog "Step 18 - 30" siclog@   Steps defined in Annex C.10

    v_InviteRequestWithSdp_Dialog3 := f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9(p_TestProcedure, v_ConferenceFactoryUri, v_TemporaryConferenceUri, v_FinalConferenceUri, v_RadioBearersEstablished);

    v_ContactUrl_UE := f_MessageHeader_GetContactSipUrl(v_InviteRequestWithSdp_Dialog3.Invite.msgHeader);

    v_MessageHeader_Bye_Dialog1 := f_IMS_ByeRequest_MessageHeaderRX(v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, -, firstDialog);

    v_MessageHeader_Bye_Dialog2 := f_IMS_ByeRequest_MessageHeaderRX(v_SessionInfo_Dialog2.InviteRequestWithSdp, -, secondDialog);

    do {

      alt {

        // @siclog "Step 27 - 30" siclog@  Annex C.10 Step 10 - 13 (UE subscribes the conference event)

        [v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed] a_ConferenceEventPackage_Subscribe(v_ConferenceEventPackage_SubscribeInfo_ByRef, v_FinalConferenceUri)

          {

            v_SubscribeRequest := v_ConferenceEventPackage_SubscribeInfo_ByRef.SubscribeRequest;

          }

        // @siclog "Step 31 - 38" siclog@  Annex C.19 Step 1 - 8

        [v_AnnexC19C37_Dialog1_ByRef.State != finished] a_IMS_InviteUserToConference_AnnexC19C37(v_AnnexC19C37_Dialog1_ByRef,

                                                                                                 v_FinalConferenceUri,

                                                                                                 firstDialog,       // @sic R5s150909: dialog index instead of boolean sic@

                                                                                                 v_SubscribeRequest,

                                                                                                 tsc_MediaTypeList_Audio,

                                                                                                 v_ConferenceEventPackage_Version)

          {

            v_WaitForBYE_Dialog1 := fl_AnnexC19C37_StateInfo_WaitForBYE(v_AnnexC19C37_Dialog1_ByRef) and not v_Dialog1_Terminated;   /* determine whether to wait for BYE for dialog 1; @sic R5s150910: v_Dialog1_Terminated sic@ */

            if (v_AnnexC19C37_Dialog1_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 1 has joined the conference ... */

              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;                 /* ... and the version needs to incremented for next notification */

            }

          }

        // @siclog "Step 39 - 46" siclog@  Annex C.19 Step 1 - 8

        [v_AnnexC19C37_Dialog2_ByRef.State != finished] a_IMS_InviteUserToConference_AnnexC19C37(v_AnnexC19C37_Dialog2_ByRef,

                                                                                                 v_FinalConferenceUri,

                                                                                                 secondDialog,      // @sic R5s150909: dialog index instead of boolean sic@

                                                                                                 v_SubscribeRequest,

                                                                                                 tsc_MediaTypeList_Audio,

                                                                                                 v_ConferenceEventPackage_Version)

          {

            v_WaitForBYE_Dialog2 := fl_AnnexC19C37_StateInfo_WaitForBYE(v_AnnexC19C37_Dialog2_ByRef) and not v_Dialog2_Terminated;   /* determine whether to wait for BYE for dialog 2; @sic R5s150910: v_Dialog2_Terminated sic@ */

            if (v_AnnexC19C37_Dialog2_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 2 has joined the conference ... */

              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;                 /* ... and the version needs to incremented for next notification */

            }

          }

        //--------------------------------------------

        // DIALOG 1

        //--------------------------------------------

        // @siclog "Step 46A - 46B" siclog@ Step numbers are missing in the prose

        [v_WaitForBYE_Dialog1] a_IMS_ReceiveByeSendOK(firstDialog, v_SessionInfo_Dialog1.ContactUrl, v_MessageHeader_Bye_Dialog1)

          {

            v_WaitForBYE_Dialog1 := false;

            v_Dialog1_Terminated := true;   /* @sic R5s150910 sic@ */

          }

        //--------------------------------------------

        // DIALOG 2

        //--------------------------------------------

        // @siclog "Step 46C - 46D" siclog@

        [v_WaitForBYE_Dialog2] a_IMS_ReceiveByeSendOK(secondDialog, v_SessionInfo_Dialog2.ContactUrl, v_MessageHeader_Bye_Dialog2)

          {

            v_WaitForBYE_Dialog2 := false;

            v_Dialog2_Terminated := true;   /* @sic R5s150910 sic@ */

          }

      }

      v_Continue := v_WaitForBYE_Dialog1 or v_WaitForBYE_Dialog2

        or (v_AnnexC19C37_Dialog1_ByRef.State != finished) or (v_AnnexC19C37_Dialog2_ByRef.State != finished)

        or (isvalue(v_SubscribeRequest) and (v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed));

    } while (v_Continue);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // @siclog "Step 47 - 48" siclog@   Terminate conference

    f_IMS_CallReleaseMT(v_InviteRequestWithSdp_Dialog3.Invite, v_ContactUrl_UE);

    f_IMS_TestBody_Set(false);

  }


Change 10 – Further re-use of test bodies for WLAN/FBBA/CatM1 test cases

In order to re-use existing TTCN implementation for WLAN/FBBA/CatM1 the test bodies of several test cases is moved to separate function which are used as common test bodies for EUTRA/WLAN/FBBA/CatM1 test cases:

a) Registration test cases

	Common Test Body
	EUTRA
	WLAN
	FBBA
	CatM1

	f_IMS_InitialRegistration
	TC_8_1
	TC_G_8_1
	TC_H_8_1
	TC_J_8_1

	f_IMS_UserInitiatedReRegistration
	TC_8_2
	n.a.
	TC_H_8_2
	n.a.

	f_IMS_MobileInitiatedDeregistration
	TC_8_3
	n.a.
	TC_H_8_3
	n.a.

	f_IMS_InvalidBehaviour_423IntervalTooBrief
	TC_8_4
	n.a.
	TC_H_8_4
	n.a.

	f_IMS_UserInitiatedReRegistration_423IntervalTooBrief
	TC_8_16
	n.a.
	TC_H_8_5
	n.a.


b) Notification test cases

	Common Test Body
	EUTRA
	WLAN
	FBBA
	CatM1

	f_IMS_NetworkInitiatedReAuthentication
	TC_11_2
	n.a.
	TC_H_11_2
	n.a.


c) Call control test cases

	Common Test Body
	EUTRA
	WLAN
	FBBA
	CatM1

	f_IMS_MO_Call_503_ServiceUnavailable
	TC_12_2
	n.a.
	TC_H_12_1
	n.a.

	f_IMS_MO_Call_503_ServerTimeout
	TC_12_2a
	n.a.
	TC_H_12_2
	n.a.

	f_IMS_MO_VoiceCall_C21C21b
	TC_12_12
	n.a.
	TC_H_12_3
	TC_J_12_1

	f_IMS_MT_VoiceCall_C11C11b
	TC_12_13
	n.a.
	TC_H_12_5
	TC_J_12_2


d) Emergency call test cases

	Common Test Body
	EUTRA
	WLAN
	FBBA
	CatM1

	f_IMS_EmergencyCallWithEmergencyRegistration_IMS1
f_IMS_EmergencyCallWithEmergencyRegistration_IMS2
	TC_19_1_1
	n.a.
	n.a.
	TC_J_19_1

	
	TC_19_1_2
	
	
	TC_J_19_2


NOTE: Further minor changes are done in common test case body functions like common use of f_IMS_TestBody_Set(true/false) within the test case body function.
Change 11 – Common functions for Media test cases

a) MO Speech, add video remove video
	  function fl_IMS_BuildSDP_MOCallAudioVideo_ReleaseVideo(SDP_Message p_OriginalSdpOffer) runs on IMS_PTC return template (present) SDP_Message

  {

    var charstring v_FmtAudio;

    var charstring v_FmtVideo;

    var template (present) SDP_attribute_rtpmap_codec v_RtpmapAttrAudio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;

    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    var template (present) SDP_attribute_list v_SDP_PrecondionAttributes;

    var template (present) SDP_Message v_SDP_MessageTemplate;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    if (v_IsFBBA) {

      v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(p_OriginalSdpOffer.media_list[0]);

      v_RtpmapAttrAudio := cr_RTPMAP_AMR_8000;

      v_SDP_PrecondionAttributes := {};

    } 

    else {

      v_FmtAudio := f_SDP_MediaDescr_GetAMR_WB_16000_1(p_OriginalSdpOffer.media_list[0]); /* @sic R5s170012/R5-171474: f_SDP_MediaDescr_GetAMR_WB_16000_1 sic@ */

      v_RtpmapAttrAudio := cr_RTPMAP_AMR_16000;   /* @sic R5s170012/R5-171474: cr_RTPMAP_AMR_16000 sic@ */

      v_SDP_PrecondionAttributes := cr_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory)); /* @sic R5s160604 change 3 sic@ */

    }

    v_FmtVideo := f_SDP_MediaDescr_GetH264_90000(p_OriginalSdpOffer.media_list[1]);

    v_SDP_MediaAttributes_Audio := {

      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_RtpmapAttrAudio),

      cr_SDP_Attribute_fmtp(v_FmtAudio)

    };

    v_SDP_MediaAttributes_Video := {

      cr_SDP_Attribute_rtpmap(v_FmtVideo, cr_RTPMAP_H264_90000),

      cr_SDP_Attribute_fmtp(v_FmtVideo)

    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,

                                        cr_SDP_Media_Audio,

                                        v_SDP_MediaAttributes_Audio,

                                        v_SDP_PrecondionAttributes,

                                        cr_SDP_Media(c_video, 0, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(v_FmtVideo)),  /* port number == 0 => release video stream

                                                                                                                        @sic R5s160604: RTP/AVPF sic@ */

                                        v_SDP_MediaAttributes_Video,

                                        v_SDP_PrecondionAttributes);

  }

	  function fl_IMS_MOCall_ReleaseVideo_Invite(SDP_Message p_OriginalSdpOffer)

runs on IMS_PTC return IMS_InviteRequestWithSdp_Type

  {

    var IMS_DATA_REQ v_IMS_INVITE_REQ;

    var SDP_Message v_SDP_Message;

    var boolean v_A11 := pc_IMS_Video_FeatureTag;

    var boolean v_A12 := false;   // as we have video but not voice

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_PreconditionSupported := not v_IsFBBA;

    var OptionTag_List v_AdditionalOptionTagsForSupported := f_IMS_AdditionalOptionTagsForSupported(v_PreconditionSupported);
    var template (present) SDP_Message v_SDP_MessageTemplate := fl_IMS_BuildSDP_MOCallAudioVideo_ReleaseVideo(p_OriginalSdpOffer);

















    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(A_2_1_A5, v_AdditionalOptionTagsForSupported, -, -, -, v_A11, v_A12);

    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_MessageTemplate);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);

  }

  

	  function f_IMS_MOSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var INVITE_Request v_InviteRequest;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;


    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;




    var template (present) SipUrl v_CalleeContactUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri);

    var charstring v_CallIdentification;
    var template (value) SDP_Message v_SdpAnswer;
    var template (value) MessageBody v_MessageBody;


    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;

    var IMS_PreconditionIndication_Type v_PreconditionIndication;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    if (v_IsFBBA) {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_AMR_only;

      v_PreconditionIndication := NoPreconditionIndication;

      v_PreconditionsRequired := false;

    } else {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR;

      v_PreconditionIndication := PreconditionRequired;

      v_PreconditionsRequired := true;

    }
    // setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, -, normalCall_ReturnId); // @sic R5s160855 sic@
    f_IMS_CC_MO_SpeechCall(v_AudioCodecList, -, v_PreconditionsRequired);

    v_CallIdentification := f_UT_ReceivedIMSCallId(MMI); // @sic R5s160855 sic@

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_UT_AddVideoToIMSCall(MMI, v_CallIdentification);

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_Video_Step2(AddVideo);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    if (v_IsFBBA) {    // G.17.1, H.17.1

      // @siclog "Step 3" siclog@

      f_IMS_MOCallSetup_AnnexC21_Step3(v_InviteRequestWithSdp);        // Send 100 Trying

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MO_AddVideo);

      // @siclog "Step 4 - 6 (H.17.1): Void" siclog@

      // @siclog "Step 4 (G.17.1)" siclog@

      // @siclog "Step 7 (H.17.1)" siclog@

      v_SdpAnswer := f_IMS_BuildSDP_MOCallAudioVideo_Step4(AddVideo, v_InviteRequestWithSdp.SdpOffer);    // same as for C.25a/C.25b

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, -, -, v_PreconditionIndication);

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response, cs_MessageBody_SDP(v_SdpAnswer))));

      // @siclog "Step 5 (G.17.1)" siclog@

      // @siclog "Step 8 (H.17.1)" siclog@

      v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

      f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactUri, v_MessageHeader_Request)));

    } 

    else {                        // 17.1
      // @siclog "Step 3 - 4" siclog@

      v_MessageHeader_Response := f_IMS_MOCallSetup_Video_Step3_4(AddVideo, v_InviteRequestWithSdp);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MO_AddVideo);

      // @siclog "Step 5 - 8" siclog@

      f_IMS_MOCallSetup_Video_Steps5_8(AddVideo, v_InviteRequestWithSdp, v_MessageHeader_Response);

      // @siclog "Step 9 - 10" siclog@

      f_IMS_MOCallSetup_AnnexC21C25_Step12_13(v_InviteRequestWithSdp);
    }
    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have really been established

    // @siclog "Step 10a" siclog@

    f_UT_IMS_RemoveVideoMedia(MMI, v_CallIdentification);

    // @siclog "Step 11" siclog@
    v_InviteRequestWithSdp := fl_IMS_MOCall_ReleaseVideo_Invite(v_InviteRequestWithSdp.SdpOffer);



    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    // @siclog "Step 12" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // @siclog "Step 12a" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 13" siclog@

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeSDP, -, -, v_PreconditionIndication);  // @sic R5s170344: cs_ContentTypeSDP, PreconditionRequired sic@


    v_MessageBody := cs_MessageBody_SDP(f_IMS_BuildSDP_CommonResponseToOfferTX(v_InviteRequestWithSdp.SdpOffer));

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response, v_MessageBody)));                     // @sic R5s170344: v_MessageBody sic@

    // @siclog "Step 14" siclog@

    v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);


    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeContactUri, v_MessageHeader_Request)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // @siclog "Step 15 - 16" siclog@

    f_IMS_CallReleaseMO(v_InviteRequest);

    f_IMS_TestBody_Set(false);


  }


b) MT Speech, add video remove video
	  function f_IMS_MTCallSetup_AnnexC11C11b(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC return IMS_CC_MT_SpeechCallInfo_Type

  { /* @sic R5-173535: new function to be used for 12.13, J.12.2 and others sic@ */

    var IMS_CC_MT_SpeechCallInfo_Type v_SpeechCallInfo;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;           /* @sic R5w140112 sic@ */

    var IMS_Response_Type v_SessionInProgress183;

    var IMS_SDP_AudioCodec_Type v_AudioCodecToBeSelected := p_AudioCodecList[0];

    // Steps 1 to 4 as defined in Annex C.11

    // @siclog "Step 1 - 4" siclog@

    v_SpeechCallInfo.InviteRequestWithSdp := f_IMS_MTCallSetup_AnnexC11_INVITE(p_AudioCodecList);

    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_SpeechCallInfo.InviteRequestWithSdp, v_AudioCodecToBeSelected);

    v_SessionInProgress183 := v_ResponseWithSdp.Response;

    v_SpeechCallInfo.ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader);

    // Establish Dedicated EPS bearer

    // See TS. 36.508 EUTRA/EPS signalling for IMS MT speech call Table 4.5A.7.3-1 Step 9-12

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  /* trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

                                                                        NOTE: Even in case of FBBA the IPCAN PTC needs to be triggered to do the routing for RTP/RTCP loopback */

    // Steps 5 to 13 as defined in Annex C.11

    // @siclog "Step 5 -13" siclog@

    f_IMS_MTCallSetup_AnnexC11_Steps5_13(v_SpeechCallInfo.InviteRequestWithSdp, v_ResponseWithSdp, v_AudioCodecToBeSelected);

    return v_SpeechCallInfo;

  }

	  function fl_IMS_ReceiveOptionalTrying_Timeout(template (value) INVITE_Request p_InviteRequest,

                                                template (present) InternetProtocol_Type p_Protocol) runs on IMS_PTC

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP_Ref;

    timer t_Timer := 5.0;    

    t_Timer.start

    alt {

      [] a_IMS_ReceiveResponse(p_InviteRequest.msgHeader,

                               p_Protocol,

                               cr_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderRX(p_InviteRequest)),

                               v_IMS_DATA_RSP_Ref);       /* @sic R5s140350: v_IMS_DATA_RSP is 'out' parameter sic@ */

      [] t_Timer.timeout;

    }

  }

	  function f_IMS_MTSpeechAddVideoRemoveVideo() runs on IMS_PTC

  {
    var IMS_CC_MT_SpeechCallInfo_Type v_SpeechCallInfo;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;

    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;

    var IMS_Response_Type v_SessionInProgress183;

    var template (value) INVITE_Request v_InviteRequest;

    var template (value)   SDP_Message v_Tx_SDP;

    var template (present) SDP_Message v_Rx_SDP;
    var template SDP_Message v_SdpAnswer200OK;
    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) MessageHeader v_MessageHeader_Update;

    var template (value) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (omit) SDP_attribute v_SDP_Attribute;

    var SipUrl v_ContactUrl;

    var InternetProtocol_Type v_Protocol;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var IMS_SDP_AudioCodecList_Type v_AudioCodecList;

    var boolean v_PreconditionsRequired;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    // preamble MT speech call setup


    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);











    if (v_IsFBBA) {

      v_AudioCodecList := tsc_IMS_SDP_AudioCodecList_AMR_only;

      v_PreconditionsRequired := false;

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11aC11c(v_AudioCodecList, v_PreconditionsRequired);

    } 

    else {

      v_SpeechCallInfo := f_IMS_MTCallSetup_AnnexC11C11b(v_AudioCodecList);

    }

    v_ContactUrl := v_SpeechCallInfo.ContactUrl;
    // MT Call is now established

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // Step 1 SS sends re-INVITE with second SDP offer to add video.

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(AddVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    if (v_IsFBBA) {

      v_Protocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

      fl_IMS_ReceiveOptionalTrying_Timeout(v_InviteRequest, v_InviteRequestWithSdp.RoutingInfo.Protocol);

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      v_SdpAnswer200OK := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo);

    }

    else {
      // @siclog "Step 2: Void" siclog@

      // @siclog "Step 3" siclog@

      v_ResponseWithSdp := f_IMS_MTCallSetup_Video_Receive183(AddVideo, v_InviteRequestWithSdp);    /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_SessionInProgress183 := v_ResponseWithSdp.Response;

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_MT_AddVideo);

      f_IMS_IPCAN_WaitForTrigger(IPCAN);

      // @siclog "Step 4 - 5" siclog@

      f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);

      // @siclog "Step 6" siclog@

      v_SDP_Attribute := f_SDP_AttributeList_GetAttribute(v_ResponseWithSdp.SdpMessage.media_list[0].attributes, cr_SDP_Attribute_curr_qos(c_local, ?));

      v_Tx_SDP := f_IMS_BuildSDP_MTCallAudioVideo_TX(AddVideo, Update, valueof(v_SDP_Attribute.curr.direction)); /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderTX(v_InviteRequest, PreconditionSupported);     /* @sic R5-160795: PreconditionSupported sic@ */

      v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();                                                    /* @sic R5s160540 change 1.5 sic@ */

      IMS_Client.send(cas_IMS_Update_Request(cs_IMS_RoutingInfo(v_Protocol), cs_UPDATE_Request(v_ContactUrl, v_MessageHeader_Update, cs_MessageBody_SDP(v_Tx_SDP))));

      // @siclog "Step 7" siclog@

      v_Rx_SDP := f_IMS_BuildSDP_MTCallAudioVideo_RX(AddVideo, Response200);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

      v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Update, cr_ContentTypeSDP);   /* @sic R5-160795, R5-162930: precondition required sic@ */

      v_IMS_DATA_RSP := f_IMS_ReceiveResponse(v_MessageHeader_Update, v_Protocol, cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));

      f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP.Response.messageBody, v_Rx_SDP);

      v_SdpAnswer200OK := omit;

    }
    // @siclog "Step 7A" siclog@

    // Step 7A Make UE accept the speech and video offer.

    f_UT_AnswerCall(MMI);

    // @siclog "Step 8 - 9" siclog@

    // Step 8 The UE responds to the re-INVITE with a 200 OK final response.

    // Step 9 The SS acknowledges the receipt of 200 OK for the re-INVITE.

    f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequestWithSdp, v_SdpAnswer200OK);

    // @siclog "Step 10" siclog@

    // Step 10 SS sends a re-INVITE with a SDP offer indicating that the video component is removed from the call

    v_InviteRequestWithSdp := f_IMS_MTCallSetup_Video_SendInvite(RemoveVideo, Invite, v_ContactUrl);  /* @sic R5-172930: v_FmtAudio, v_FmtVideo removed sic@ */

    // @siclog "Step 10a - 11" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);    /* @sic R5-160795, R5-162930: cr_ContentTypeSDP, precondition required sic@ */

    f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite,

                                         v_InviteRequestWithSdp.RoutingInfo.Protocol,

                                         cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP));  /* @sic R5s160842, R5s170013 sic@ */

    // @siclog "Step 11a" siclog@

    f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_ReleaseVideo);

    // @siclog "Step 12" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrl, v_MessageHeader_Ack)));

    f_IMS_IPCAN_WaitForTrigger(IPCAN);    // Wait for trigger from IPCAN to ensure that radio resources have been released

    // Step 13 void // @sic R5-134646 sic@

    // Step 14 and 15. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl);

    f_IMS_TestBody_Set(false);


  }

	  function f_TC_17_2_IMS1() runs on IMS_PTC

  {

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

    f_IMS_MTSpeechAddVideoRemoveVideo();

    f_IMS_CC_Postamble(IPCAN_MT_SpeechCall);

  }


